

Az AB függvények

Az AB függvények esetén a feltételek megadása – nevüknek megfelelően – az adatbázis-kezelésben használt lekérdezésekre hasonlít. Az AB függvények már jóval a HATÖBB függvények előtt is szerepeltek az Excelben.

Az AB függvények esetén a cég dolgozóinak adatait, vagyis az A1:G31 tartományt **adatbázisnak** nevezik. Az adatbázis oszlopait **mezőknek** hívják, a mezőket a **mezőnév** azonosítja, amely az adatbázis első sorában található. Gyakran használják még az adatbázis adatsoraira a **rekord** elnevezést is.

Első példánkban a *beszerzés* részlegen dolgozó *vidékiek* átlagfizetését számítjuk ki, ehhez az AB.ÁTLAG függvényt fogjuk használni.

Az AB függvények esetében a feltételek egy táblázatba kerülnek, ezt a táblázatot *kritériumtartománynak* nevezik. A következő ábrán az I1:J2 tartományban találjuk a kritériumot. A kritériumtartomány első sorába a megfelelő mezőnevek, közvetlen alájuk pedig a feltételek kerülnek. Az egymás mellé kerülő feltételeket az ÉS művelet köti össze. Példánkban a kritérium: Részleg=beszerzés ÉS Város<>Budapest. (A szöveget nem kell idézőjelek közé tenni.)

Az AB függvényeknek három paraméterük van: az első az adatbázis, a második a számításokban használt mező azonosítója és a harmadik a kritériumtartomány. Esetünkben tehát a beszerzésen dolgozó vidékiek átlagfizetését a *Fizetés* mező felhasználásával a következő függvény adja meg:

=AB.ÁTLAG(A1:G32;E1;I1:J2)

	C	D	E	F	G	H	I	J	K	L
1	Város	Szülev	Fizetés	Jutalom	Neme		Részleg	Város		
2	Kotkodács	1956	549 835	80 000	nő		beszerzés	<>Budapest		
3	Budapest	1969	370 806	80 000	férfi					
4	Tótújfalu	1984	356 709	10 000	férfi		Átlagfizetés:	=AB.ÁTLAG(A1:G32;E1;I1:J2)		
5	Budapest	1970	474 062	30 000	nő					

Az AB függvényeknek három paraméterük van: az adatbázis (A1:G32), a mező neve, amelyre a művelet vonatkozik (E1) és a feltételeket megadó kritériumtartomány (I1:J2)

Az AB függvények elnevezése megfelel a már megszokott statisztikai függvényeknek. Így egy adott feltételnek megfelelő adathalmaz összegét, átlagát, maximumát, minimumát, szórását, elemszámát rendre az AB.SZUM, AB.ÁTLAG, AB.MAX, AB.MIN, AB.SZÓRÁS, AB.DARAB függvények adják. Az AB.DARAB mellett (ami a kijelölt tartományban lévő számokat számolja meg) ezúttal is létezik az AB.DARAB2 függvény is, ami a nem üres cellák számát adja. Az AB függvények alkalmazására a következő ábrán láthatunk példákat.

Feltételhez kötött számítások

A mezőnevek megadása többféleképpen történhet, ahogy ezt az ábrán is láthatjuk. Szerepelhet a mezőnevet tartalmazó cella címe (E1), maga a mezőnév („Fizetés”) vagy a mező sorszáma (5) egyaránt.

	E	F	G	H	I	J	K	L	M
1	Fizetés	Jutalom	Neme		Részleg	Város			
2	549 835	80 000	nő		beszerzés	<>Budapest			
3	370 806	80 000	férfi						
4	356 709	10 000	férfi		Átlagfizetés:	268349	=AB.ÁTLAG(A1:G31;E1;I1:J2)		
5	474 062	30 000	nő		Bértömeg:	2146793	=AB.SZUM(A1:G31;"Fizetés";I1:J2)		
6	328 710	80 000	férfi		Legnagyobb bér:	549835	=AB.MAX(A1:G31;5;I1:J2)		
7	83 267	50 000	férfi	Legöregebb születési éve:		1926	=AB.MIN(A1:G31;D1;I1:J2)		
8	374 518	10 000	férfi	Dolgozók száma(1):		8	=AB.DARAB(A1:G31;D1;I1:J2)		
9	180 144	10 000	férfi	Dolgozók száma(2):		8	=AB.DARAB2(A1:G31;A1;I1:J2)		

Példák az AB függvények alkalmazására. A mezőnevet megadhatjuk a cella címével (pl. E1), a mezőnév beírásával (pl. „Jutalom”) vagy a mező sorszámaival (5) is

Egy további gyakran használt AB függvény az AB.MEZŐ, amely visszaadja egy kiválasztott mezőből a kritériumoknak megfelelő értéket. Segítségével megkereshetjük például a beszerzésen dolgozó vidékiek közül a legöregebb nevét.

A megoldáshoz először a kritériumtartományt még egy feltétellel bővítenünk kell, ez a feltétel a legöregebb dolgozó születési évét tartalmazza. Tekintsük a következő ábrát! Az új feltétel a Szülév mezőre vonatkozik, amelynek kívánt értékét az előző példából ismert AB.MIN(A1:G31;"Szülév";I1:J2) függvény segítségével számítjuk ki (K2 cella). Ezúttal tehát már a feltételt is képlettel kapjuk.

Az összetett feltételt a bővített I1:K2 kritériumtartomány tartalmazza, innen a dolgozó nevét pedig az AB.MEZŐ függvénnyel megkaphatjuk:

=AB.MEZŐ(A1:G31;"Név";I1:K2)

	H	I	J	K	L	M	N
1		Részleg	Város	Szülév			
2		beszerzés	<>Budapest	1926	=AB.MIN(A1:G31;"Szülév";I1:J2)		
3							
4	Legöregebb dolgozó:	Czeczei Zsolt		=AB.MEZŐ(A1:G31;"Név";I1:K2)			

A beszerzésen dolgozó legöregebb vidéki kolléga nevének megkeresése. A dolgozó születési évét az AB.MIN függvény, ezt felhasználva a nevét az AB.MEZŐ függvény adja vissza

Az AB.MEZŐ csak akkor működik, ha a kritériumoknak pontosan egy érték felel meg az adott mezőben, egyébként a #SZÁM! hibaüzenetet kapjuk.

A kritérium megadásánál gyakori hiba, hogy elírjuk a mezőnevet (tipikus probléma például, hogy a mezőnév után van egy szóköz, amit nem veszünk észre). Ezért a mezőneveket vagy a vágólapon át érdemes átmásolni, vagy hivatkozással célszerű szerepeltetni. Most például a J1 cella tartalma az =C1 képlet.